
Use lists on screens in your programs.

At the conclusion of this unit, you will be able to:

Lists in Screen Programming: Unit Objectives

· Fill the corresponding basic list buffer with WRITE statements at PBO or PAI. You can create your own list and column headers by programming an event TOP-OF-PAGE. This event will be triggered whenever a new page is created in the list buffer with NEW-PAGE.

· You can send the output directly to the spool with the NEW-PAGE PRINT ON  statement. 

[image: image5.wmf]ã

SAP AG 1999

Screen 300

PBO/PAI (Screen 100)

LEAVE TO LIST

-

PROCESSING

PBO/PAI (Screen 300)

LEAVE TO LIST

-

PROCESSING

CALL SCREEN 200.

SET SCREEN 300.

List Display at the Front End

Screen 100

Screen 200


· There is no common list buffer outside of a CALL level. 

· The list display is processed at the end of the screen in which LEAVE TO LIST-PROCESSING was programmed at PBO or PAI. 

· To direct the output to the spool, use the NEW-PAGE PRINT ON statement, but not LEAVE TO LIST-PROCESSING.

· To create a list that is displayed on a screen, use the ABAP statement LEAVE TO LIST-PROCESSING. This sets a switch that ensures that the contents of the list buffer are output once the current screen has been processed. The SET PF-STATUS SPACE statement ensures that the list is displayed with the standard GUI status for lists.

· Once the screen has been fully processed and LEAVE TO LIST- PROCESSING is executed, the list is displayed on list screen 120 (screen for a basis program).

· You can also use the following format: LEAVE TO LIST-PROCESSING AND RETURN TO SCREEN 0. SET PF-STATUS SPACE. WRITE ...  LEAVE SCREEN. 

· When the system exits list processing (user selects F3 or ABAP statement LEAVE LIST-PROCESSING), the system carries on processing the PBO of the calling screen from which the list processing was started. You can override this by using the AND RETURN TO SCREEN <scr> addition in the LEAVE TO LIST-PROCESSING statement.
· If you include the ABAP statement SUPPRESS DIALOG in a PBO module, the current screen is not displayed.  

· If you want to display a list in a dialog box within a transaction, you must call a screen, but include the SUPPRESS DIALOG statement in its PBO processing block.

· In order to return to the calling screen when the user leaves the list, use the statement
LEAVE TO LIST-PROCESSING AND RETURN TO SCREEN 0.

Use lists on screens in your programs

You are now able to:

Lists in Screen Programming: Unit Summary 


Lists in Screen Programming Optional Exercises 

	Unit: Lists in Screen Programming

Topic: Displaying a list on a screen

	At the conclusion of these exercises, you will be able to:

· Use lists on screens in your programs.

	Extend your flight maintenance screen to display a booking list. Allow the user to display the bookings by choosing a pushbutton or menu entry.  


Program the booking list. 

1-1
Extend your program SAPMZ##BC410_SOLUTION from the previous exercise (or copy the model solution SAPMBC410ACONS_CONTEXTMENU). Use the model solution SAPMBC410ALISS_LISTS_ON_DYNPROS for orientation.

1-2
Write a subroutine display_bookings to display the data from the internal table it_sbook. Display the following data: 

Booking number (bookid), 
Customer number (customid) 
Customer type (custtype) 
Luggage weight (luggweight) 
Unit of weight (wunit) 
Class (class) 
Booking date (order_date).

1-3
Assign the function BOOK in status STATUS_100 to function key F5, and also to a pushbutton and a menu entry.  

1-4
Call screen 200 (STARTING AT…) if the user chooses the BOOK function.  Create the screen (type: Modal dialog box). In the PBO event of the screen, call a module in which you create the list.  Screen 200 is only a container. You should not actually display it. Create a GUI title (T_BOOK) for the booking list. Set the GUI status SPACE and GUI title T_BOOK and start list processing.

Lists in Screen Programming Solutions

	Unit: Lists in Screen Programming

Topic: Displaying a list on a screen


Model solution SAPMBC410ALISS_LISTS_ON_DYNPROS
Add the coding in bold type to your program and create the module. 

Flow logic screen 200

PROCESS BEFORE OUTPUT.

  MODULE list.

*

PROCESS AFTER INPUT.

PBO module include

Add the coding in bold type to your include

MODULE status_0100 OUTPUT.

  SET PF-STATUS 'STATUS_100'.

  SET TITLEBAR 'TITLE_100'.

ENDMODULE.                             " STATUS_0100  OUTPUT

MODULE clear_ok_code OUTPUT.

  CLEAR ok_code.

ENDMODULE.                             " clear_ok_code  OUTPUT

MODULE modify_screen OUTPUT.

  CHECK NOT mode-maintain_flights IS INITIAL.

  LOOP AT SCREEN.

    IF screen-name = 'SDYN_CONN-PLANETYPE'.

      screen-input = 1.

      screen-required = 1.

      MODIFY SCREEN.

    ENDIF.

  ENDLOOP.

ENDMODULE.                             " modify_screen  OUTPUT

MODULE get_spfli OUTPUT.

  ON CHANGE OF wa_sflight-carrid OR wa_sflight-connid.

    SELECT SINGLE * INTO CORRESPONDING FIELDS OF sdyn_conn FROM spfli

      WHERE carrid = wa_sflight-carrid

        AND connid = wa_sflight-connid.

  ENDON.

ENDMODULE.                             " GET_SPFLI  OUTPUT

MODULE get_saplane OUTPUT.

  ON CHANGE OF wa_sflight-planetype.

    SELECT SINGLE * FROM saplane WHERE planetype = wa_sflight-planetype.

  ENDON.

ENDMODULE.                             " GET_SAPLANE  OUTPUT

MODULE fill_dynnr OUTPUT.

  CASE my_tabstrip-activetab.

    WHEN 'FC1'.

      dynnr = 110.

    WHEN 'FC2'.

      dynnr = 120.

    WHEN 'FC3'.

      dynnr = 130.

    WHEN OTHERS.

      my_tabstrip-activetab = 'FC1'.

      dynnr = 110.

  ENDCASE.

ENDMODULE.                             " set_dynnr  OUTPUT

MODULE get_sbook OUTPUT.

  IF wa_sflight-carrid <> sdyn_book-carrid OR

     wa_sflight-connid <> sdyn_book-connid OR

     wa_sflight-fldate <> sdyn_book-fldate OR

     bookings_changed = 'X'.

    CLEAR bookings_changed.

    SELECT * FROM sbook INTO CORRESPONDING FIELDS OF TABLE it_sdyn_book

        WHERE carrid = wa_sflight-carrid

          AND connid = wa_sflight-connid

          AND fldate = wa_sflight-fldate

          AND cancelled = not_cancelled.

    DESCRIBE TABLE it_sdyn_book LINES my_table_control-lines.

  ENDIF.

ENDMODULE.                             " get_sbook  OUTPUT

MODULE trans_to_tc OUTPUT.

  MOVE wa_sdyn_book TO sdyn_book.

ENDMODULE.                             " trans_to_dynp  OUTPUT

MODULE modify_screen_130 OUTPUT.

  CHECK NOT mode-maintain_bookings IS INITIAL.

  LOOP AT SCREEN.

    IF screen-name = 'P_DELETE'.

      screen-invisible = 0.

      MODIFY SCREEN.

    ENDIF.

  ENDLOOP.

ENDMODULE.                             " modify_screen_130  OUTPUT
MODULE list OUTPUT.
  SELECT * FROM sbook

      INTO TABLE it_sbook

      WHERE carrid = wa_sflight-carrid

        AND connid = wa_sflight-connid

        AND fldate = wa_sflight-fldate

        AND cancelled = space.

  IF sy-dbcnt = 0.

    MESSAGE i186.

  ELSE.

    PERFORM display_bookings.

    SET PF-STATUS space.

    SET TITLEBAR 'T_BOOK'.

  ENDIF.

  LEAVE TO LIST-PROCESSING AND RETURN TO SCREEN 0.

  SUPPRESS DIALOG.
ENDMODULE.                             " LIST  OUTPUT

Subroutines include

Add the coding in bold type to your include

FORM update_sflight.

  UPDATE sflight FROM wa_sflight.

  IF sy-subrc NE 0.

    MESSAGE a008.

*   Fehler beim Ändern

  ENDIF.

  MESSAGE s009.

ENDFORM.                               " update_sflight

FORM cancel_bookings.

  DATA: wa_sbook_upd TYPE sbook,

        it_sbook_upd LIKE TABLE OF wa_sbook_upd.

  CONSTANTS cancelled VALUE 'X'.

  LOOP AT it_sdyn_book INTO wa_sdyn_book WHERE mark = 'X'.

    MOVE-CORRESPONDING wa_sdyn_book TO wa_sbook_upd.

    MOVE cancelled TO wa_sbook_upd-cancelled.

    APPEND wa_sbook_upd TO it_sbook_upd.

  ENDLOOP.

  CALL FUNCTION 'BC_GLOBAL_UPDATE_BOOK'

       TABLES

            booking_tab     = it_sdyn_book

            booking_tab_upd = it_sbook_upd.

  MESSAGE s009.

ENDFORM.                               " cancel_bookings

FORM on_ctmenu_sub130 USING p_menu TYPE REF TO cl_ctmenu.

  DATA fcodes TYPE ui_functions.

  CALL METHOD cl_ctmenu=>load_gui_status

            EXPORTING program = sy-cprog

                      status = 'S_SUB130'

                      menu = p_menu.

  CHECK NOT mode-maintain_bookings IS INITIAL.

  APPEND 'DELE' TO fcodes.

  CALL METHOD p_menu->enable_functions EXPORTING fcodes = fcodes.

ENDFORM.                               " on_ctmenu_sub130

FORM display_bookings.

  LOOP AT it_sbook INTO wa_sbook.

    WRITE: / wa_sbook-bookid,

             wa_sbook-customid,

             wa_sbook-custtype,

             wa_sbook-luggweight UNIT wa_sbook-wunit,

             wa_sbook-wunit,

             wa_sbook-class,

             wa_sbook-order_date,

             wa_sbook-cancelled.

  ENDLOOP.

ENDFORM.                               " DISPLAY_BOOKINGS

© SAP AG
TAW12
10-13

